Pensieve header: A $V\times F$ formula for the Alexander polynomial following Kauffman's Formal Knot Theory, written by DBN in February 2024. Almost exactly matches with the Martchenkov program.

```
In[•]:= << KnotTheory`
```

```
Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
```

```
In[•]:= K = Knot[3, 1]
Out[•]= Knot[3, 1]
```

```
In[•]:= n = Length[pd = List @@ PD[K]]
```

KnotTheory: Loading precomputed data in PD4Knots`.

```
Out[•]= 3
```

```
In[•]:= pd
Out[•]= {X[1, 4, 2, 5], X[3, 6, 4, 1], X[5, 2, 6, 3]}
```

```
In[•]:= XingsByArmpits =
         pd /. x : X[i_, j_, k_, l_] :→ If[PositiveQ[x], X+[-i, j, k, -l], X-[-j, k, l, -i]]
Out[•]= {X-[-4, 2, 5, -1], X-[-6, 4, 1, -3], X-[-2, 6, 3, -5]}
```

```
In[•]:= bends = Times @@ XingsByArmpits /. _[X][a_, b_, c_, d_] :→ p[a,-d] p[b,-a] p[c,-b] p[d,-c]
```

Out[•]= $p_{-6,3}\, p_{-5,-3}\, p_{-4,1}\, p_{-3,-1}\, p_{-2,5}\, p_{-1,-5}\, p_{1,-4}\, p_{2,4}\, p_{3,-6}\, p_{4,6}\, p_{5,-2}\, p_{6,2}$

```
In[•]:= faces = bends //. p[x__,y_] p[y_,z__] :→ p[x,y,z]
```

Out[•]= $p_{-6,3,-6}\, p_{-4,1,-4}\, p_{-2,5,-2}\, p_{-1,-5,-3,-1}\, p_{6,2,4,6}$

```
In[•]:= M = SparseArray[Flatten@Table[
           x = XingsByArmpits[[i]];
           js = Position[faces, #][[1, 1]] & /@ List @@ x;
           If[Head[x] === X+, s = 1, s = -1];
           {{i, js[[1]]} → T^-s, {i, js[[2]]} → 1, {i, js[[3]]} → T^s, {i, js[[4]]} → 1},
           {i, n}], {n, n + 2}]
Out[•]= SparseArray[ Specified elements: 12  Dimensions: {3, 5} ]
```

```
In[•]:= Normal[M]
```

Out[•]= $\left\{\left\{0, T, \frac{1}{T}, 1, 1\right\}, \left\{T, \frac{1}{T}, 0, 1, 1\right\}, \left\{\frac{1}{T}, 0, T, 1, 1\right\}\right\}$

In[∘]:=
```
Det[Delete[{Position[faces, #][[1, 1]]} & /@ {1, -1}] /@ Normal[M]] /. T → T^(1/2)
```

Out[∘]=

$$\frac{1 - T + T^2}{T}$$

In[∘]:=
```
A[K_] := Module[{pd, n, XingsByArmpits, bends, faces, M, x, js, i},
  n = Length[pd = List @@ PD[K]];
  XingsByArmpits =
   pd /. x : X[i_, j_, k_, l_] ⧴ If[PositiveQ[x], X_+[-i, j, k, -l], X_-[-j, k, l, -i]];
  bends = Times @@ XingsByArmpits /. _[X][a_, b_, c_, d_] ⧴ p_{a,-d} p_{b,-a} p_{c,-b} p_{d,-c};
  faces = bends //. p_{x__,y_} p_{y_,z__} ⧴ p_{x,y,z};
  M = SparseArray[Flatten@Table[
      x = XingsByArmpits[[i]];
      js = Position[faces, #][[1, 1]] & /@ List @@ x;
      If[Head[x] === X_+, s = 1, s = -1];
      {{i, js[[1]]} → T^-s, {i, js[[2]]} → 1, {i, js[[3]]} → T^s, {i, js[[4]]} → 1},
      {i, n}], {n, n + 2}];
  Expand[Det[Delete[{Position[faces, #][[1, 1]]} & /@ {1, -1}] /@ Normal[M]]
   ] /. T → T^(1/2)
 ]
```

In[∘]:=
```
A /@ AllKnots[{3, 7}]
```

Out[∘]=

$$\left\{-1 + \frac{1}{T} + T,\ -3 + \frac{1}{T} + T,\ 1 + \frac{1}{T^2} - \frac{1}{T} - T + T^2,\ 3 - \frac{2}{T} - 2T,\ 5 - \frac{2}{T} - 2T,\ 3 + \frac{1}{T^2} - \frac{3}{T} - 3T + T^2,\right.$$
$$-5 - \frac{1}{T^2} + \frac{3}{T} + 3T - T^2,\ -1 + \frac{1}{T^3} - \frac{1}{T^2} + \frac{1}{T} + T - T^2 + T^3,\ 5 - \frac{3}{T} - 3T,\ -3 - \frac{2}{T^2} + \frac{3}{T} + 3T - 2T^2,$$
$$\left.-7 + \frac{4}{T} + 4T,\ -5 - \frac{2}{T^2} + \frac{4}{T} + 4T - 2T^2,\ -7 - \frac{1}{T^2} + \frac{5}{T} + 5T - T^2,\ 9 + \frac{1}{T^2} - \frac{5}{T} - 5T + T^2\right\}$$

In[∘]:=
```
Table[Factor[A[K] / Alexander[K][T]], {K, AllKnots[{3, 10}]}]
```

Out[∘]=
```
{1, -1, 1, -1, 1, -1, -1, 1, -1, -1, 1, -1, 1, 1, -1, -1, -1, -1, 1, -1, -1, 1, -1, 1, -1, 1,
 -1, 1, 1, -1, -1, 1, 1, 1, 1, 1, 1, -1, -1, -1, 1, 1, -1, 1, 1, -1, -1, 1, 1, -1, -1, 1,
 -1, 1, -1, -1, -1, -1, -1, 1, -1, -1, 1, 1, 1, -1, 1, -1, -1, 1, 1, 1, -1, 1, -1, 1, -1,
 1, 1, 1, 1, -1, 1, 1, -1, -1, 1, 1, -1, -1, -1, -1, 1, -1, -1, 1, -1, -1, -1, -1, -1,
 -1, -1, 1, 1, -1, -1, 1, 1, -1, 1, 1, -1, -1, -1, -1, -1, -1, 1, 1, -1, -1, -1, 1, -1,
 -1, 1, -1, 1, 1, 1, -1, -1, 1, 1, -1, 1, -1, -1, -1, 1, -1, 1, -1, 1, -1, -1, 1, 1, -1,
 -1, 1, 1, 1, 1, -1, -1, 1, 1, 1, -1, 1, -1, -1, -1, -1, -1, -1, -1, 1, -1, -1, 1, -1,
 -1, -1, -1, -1, 1, -1, 1, 1, 1, -1, -1, -1, 1, 1, -1, 1, 1, 1, -1, -1, 1, 1, -1, 1, -1,
 -1, 1, 1, 1, 1, -1, -1, -1, 1, -1, -1, -1, 1, 1, 1, 1, -1, 1, 1, 1, 1, -1, -1, -1, 1,
 -1, 1, -1, 1, 1, -1, -1, -1, -1, -1, -1, -1, -1, -1, 1, 1, 1, 1, 1, 1, 1, -1, 1, -1, 1}
```